Math 277
Ordinary Differential Equations

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 10

The Superposition Principle and Initial Value Problems

Initializations

```
> restart;
```

10.1 The Superposition Principle and Initial Value Problems

The superposition principle states that if y_1 is a solution of

$$ay'' + b\,y' + cy = f_1(t)$$

and y_2 is a solution of

$$ay'' + b\,y' + cy = f_2(t)$$

then $y_1 + y_2$ is a solution of

$$ay'' + b\,y' + cy = f_1(t) + f_2(t)$$

The proof is simple and will be presented in class. We can use the superposition principle together with the method of undetermined coefficients to solve an entire class of frequently occurring initial value problems.

Examples

Example 10.1.1

Solve the initial value problem

$$y'' + y' - 2\,y = 3\,t + 2e^t \qquad y(0) = -3, y'(0) = 4$$

Solution

First we determine the roots of the auxiliary equation of the corresponding homogeneous equation.

```
> aux:=r^2+r-2=0;
  ev:=solve(aux, r);
```

$$aux := r^2 + r - 2 = 0$$

$$ev := 1, -2 \qquad \mathbf{(2.1.1.1)}$$

This means that the general solution of the homogeneous equation is given by

```
> ygh:=add(c[k]*exp(ev[k]*t), k=1..2);
```

$$ygh := c_1\,e^t + c_2\,e^{-2\,t} \qquad \mathbf{(2.1.1.2)}$$

Since 1 is a root of the auxiliary equation, we try

$$y_p(t) = A_0 + A_1\,t + A_2\,t\,e^t$$

as a particular solution of the inhomogeneous equation.

```
> deq:=diff(y(t), t$2)+diff(y(t), t)-2*y(t)=3*t+2*exp(t);
```

```
ytry:=A[0]+A[1]*t+A[2]*t*exp(t);
```

$$deq := \frac{d^2}{dt^2}\, y(t) + \frac{d}{dt}\, y(t) - 2\, y(t) = 3\, t + 2\, e^t$$

$$ytry := A_0 + A_1\, t + A_2\, t\, e^t \qquad \textbf{(2.1.1.3)}$$

Substitute the trial solution into the differential equation and compare the coefficients of likewise terms.

```
> eq:=eval(deq, y(t)=ytry);
```

$$eq := 3\, A_2\, e^t + A_1 - 2\, A_0 - 2\, A_1\, t = 3\, t + 2\, e^t \qquad \textbf{(2.1.1.4)}$$

```
> pars:=solve(identity(eq, t), {seq(A[k], k=0..2)});
```

$$pars := \left\{A_0 = -\frac{3}{4},\, A_1 = -\frac{3}{2},\, A_2 = \frac{2}{3}\right\} \qquad \textbf{(2.1.1.5)}$$

A particular solution is given by

```
> yp:=eval(ytry, pars);
```

$$yp := -\frac{3}{4} - \frac{3}{2}\, t + \frac{2}{3}\, t\, e^t \qquad \textbf{(2.1.1.6)}$$

We can now assemble the general solution of the inhomogeneous equation as the sum of the general solution of the homogeneous equation and a particular solution of the inhomogeneous equation.

```
> yg:=ygh+yp;
```

$$yg := c_1\, e^t + c_2\, e^{-2\,t} - \frac{3}{4} - \frac{3}{2}\, t + \frac{2}{3}\, t\, e^t \qquad \textbf{(2.1.1.7)}$$

The constants c_1 and c_2 are determined by the initial conditions.

```
> eq1:=eval(yg, t=0)=-3;
  eq2:=eval(diff(yg, t), t=0)=4;
  val_c:=solve({eq1, eq2}, {c[1], c[2]});
```

$$eq1 := c_1 + c_2 - \frac{3}{4} = -3$$

$$eq2 := c_1 - 2\, c_2 - \frac{5}{6} = 4$$

$$val\_c := \left\{c_1 = \frac{1}{9},\, c_2 = -\frac{85}{36}\right\} \qquad \textbf{(2.1.1.8)}$$

The solution of the stated initial value problem is given by

```
> sol:=eval(yg, val_c);
```

$$sol := \frac{1}{9}\, e^t - \frac{85}{36}\, e^{-2\,t} - \frac{3}{4} - \frac{3}{2}\, t + \frac{2}{3}\, t\, e^t \qquad \textbf{(2.1.1.9)}$$

Example 10.1.2

The solution of some innocent looking initial value problems may involve a hefty dose of arithmetic, as is shown by the following example

$$2\, y'' + 3\, y' + 5\, y = te^{-t} + 2e^{-3\,t} \cos 5\, t \qquad y(0) = 3,\, y'(0) = -1$$

Solution

First we determine the roots of the auxiliary equation of the corresponding homogeneous equation.

```
> aux:=2*r^2+3*r+5=0;
  ev:=solve(aux, r);
```

$$aux := 2\,r^2 + 3\,r + 5 = 0$$

$$ev := -\frac{3}{4} + \frac{1}{4}\,\mathrm{I}\sqrt{31},\ -\frac{3}{4} - \frac{1}{4}\,\mathrm{I}\sqrt{31} \quad (2.1.2.1)$$

This means that the general solution of the homogeneous equation is given by

```
> ygh:=c[1]*exp(Re(ev[1])*t)*cos(Im(ev[1])*t)+c[2]*exp(Re(ev[1])*t)*sin(Im(ev[1])*t);
```

$$ygh := c_1\,\mathrm{e}^{-\frac{3}{4}t}\cos\left(\frac{1}{4}\sqrt{31}\,t\right) + c_2\,\mathrm{e}^{-\frac{3}{4}t}\sin\left(\frac{1}{4}\sqrt{31}\,t\right) \quad (2.1.2.2)$$

Since the right hand side of the differential equation contains no solutions of the corresponding homogeneous equation, we can try a particular solution of the form

$$y_p(t) = \left(A_0 + A_1\,t\right)\mathrm{e}^{-t} + A_3\,\mathrm{e}^{-3t}\cos(5\,t) + A_4\,\mathrm{e}^{-3t}\sin(5\,t)$$

```
> deq:=2*diff(y(t), t$2)+3*diff(y(t), t)+5*y(t)=t*exp(-t)+2*exp(-3*t)*cos(5*t);
  ytry:=(A[0]+A[1]*t)*exp(-t)+A[3]*exp(-3*t)*cos(5*t)+A[4]*exp(-3*t)*sin(5*t);
```

$$deq := 2\left(\frac{\mathrm{d}^2}{\mathrm{d}t^2}\,y(t)\right) + 3\left(\frac{\mathrm{d}}{\mathrm{d}t}\,y(t)\right) + 5\,y(t) = t\,\mathrm{e}^{-t} + 2\,\mathrm{e}^{-3t}\cos(5\,t)$$

$$ytry := \left(A_0 + A_1\,t\right)\mathrm{e}^{-t} + A_3\,\mathrm{e}^{-3t}\cos(5\,t) + A_4\,\mathrm{e}^{-3t}\sin(5\,t) \quad (2.1.2.3)$$

Substitute the trial solution into the differential equation and compare the coefficients of likewise terms.

```
> eq:=eval(deq, y(t)=ytry);
```

$$eq := -A_1\,\mathrm{e}^{-t} + 4\left(A_0 + A_1\,t\right)\mathrm{e}^{-t} - 36\,A_3\,\mathrm{e}^{-3t}\cos(5\,t) + 45\,A_3\,\mathrm{e}^{-3t}\sin(5\,t) - 36\,A_4\,\mathrm{e}^{-3t}\sin(5\,t) - 45\,A_4\,\mathrm{e}^{-3t}\cos(5\,t) = t\,\mathrm{e}^{-t} + 2\,\mathrm{e}^{-3t}\cos(5\,t) \quad (2.1.2.4)$$

```
> pars:=solve(identity(eq, t), {seq(A[k], k=0..4)});
```

$$pars := \left\{A_0 = \frac{1}{16}, A_1 = \frac{1}{4}, A_2 = A_2, A_3 = -\frac{8}{369}, A_4 = -\frac{10}{369}\right\} \quad (2.1.2.5)$$

A particular solution is given by

```
> yp:=eval(ytry, pars);
```

$$yp := \left(\frac{1}{16} + \frac{1}{4}\,t\right)\mathrm{e}^{-t} - \frac{8}{369}\,\mathrm{e}^{-3t}\cos(5\,t) - \frac{10}{369}\,\mathrm{e}^{-3t}\sin(5\,t) \quad (2.1.2.6)$$

We can now assemble the general solution of the inhomogeneous equation as the sum of the general solution of the homogeneous equation and a particular solution of the inhomogeneous equation.

```
> yg:=ygh+yp;
```

(2.1.2.7)

$$yg := c_1 e^{-\frac{3}{4}t} \cos\left(\frac{1}{4}\sqrt{31}\,t\right) + c_2 e^{-\frac{3}{4}t} \sin\left(\frac{1}{4}\sqrt{31}\,t\right) + \left(\frac{1}{16} + \frac{1}{4}t\right) e^{-t} - \frac{8}{369} e^{-3t}\cos(5t) - \frac{10}{369} e^{-3t}\sin(5t) \quad \textbf{(2.1.2.7)}$$

The constants c_1 and c_2 are determined by the initial conditions.

```
> eq1:=eval(yg, t=0)=3;
  eq2:=eval(diff(yg, t), t=0)=-1;
  val_c:=solve({eq1, eq2}, {c[1], c[2]});
```

$$eq1 := c_1 + \frac{241}{5904} = 3$$

$$eq2 := -\frac{3}{4}c_1 + \frac{691}{5904} + \frac{1}{4}c_2\sqrt{31} = -1$$

$$val\_c := \left\{c_1 = \frac{17471}{5904}, c_2 = \frac{26033}{183024}\sqrt{31}\right\} \quad \textbf{(2.1.2.8)}$$

The solution of the stated initial value problem is given by

```
> sol:=eval(yg, val_c);
```

$$sol := \frac{17471}{5904} e^{-\frac{3}{4}t} \cos\left(\frac{1}{4}\sqrt{31}\,t\right) + \frac{26033}{183024}\sqrt{31}\, e^{-\frac{3}{4}t} \sin\left(\frac{1}{4}\sqrt{31}\,t\right) + \left(\frac{1}{16} + \frac{1}{4}t\right) e^{-t} - \frac{8}{369} e^{-3t}\cos(5t) - \frac{10}{369} e^{-3t}\sin(5t) \quad \textbf{(2.1.2.9)}$$

There are some impressive rationals in this expression and even the use of decimals does not make it much easier to look at.

```
> solf:=evalf(sol, 4);
```

$$solf := 2.959\, e^{-0.7500\,t}\cos(1.392\,t) + 0.7918\, e^{-0.7500\,t}\sin(1.392\,t) + (0.06250 + 0.2500\,t)\, e^{-1.\,t} - 0.02168\, e^{-3.\,t}\cos(5.\,t) - 0.02710\, e^{-3.\,t}\sin(5.\,t) \quad \textbf{(2.1.2.10)}$$

Often a mixed approach, where only the coefficients are written in decimal form, gives the best result. For instance, again in four-digit arithmetic, the solution of this problem can be written as

$$y(t) = 2.959\, e^{-\frac{3}{4}t} \cos\left(\frac{1}{4}\sqrt{31}\,t\right) + 0.7918\, e^{-\frac{3}{4}t} \sin\left(\frac{1}{4}\sqrt{31}\,t\right) + (0.06250 + 0.2500\,t)\, e^{-t} - 0.02168\, e^{-3t}\cos(5t) - 0.02710\, e^{-3t}\sin(5t)$$